Пишу запрос.
В отчет попадают записи о всех заказах, где есть указанное количество заказанных товаров и реализованных.
Хочу теперь из отчета убрать все те заказы, если количество заказано = реализовано.
Но при этом убираются эти записи при включении фильтра, "Если всё отгружено" (реализовываю как параметр да/нет).
Где можно в СКД записать функцию отсева? в Вычисляемом поле не воспринимает.
---
т на форме (скрин) если я отмечаю "ОтсевОтгружденых" - "Да" (Истина).
то те заказы, в которых Остаток "Ноль" - исчезают из отчета.

Так мой вариант и предполагает что если ОтсевОтгруженных = Да то мы убираем из запроса строки где количество из заказа = количество реализовано
---
Можно в сам запрос включить:

```1c
    ВЫБОР КОГДА &УбратьРеализованные ТОГДА
    Таблица.Заказано <> Таблица.Реализовано
    ИНАЧЕ
    ИСТИНА
    КОНЕЦ
```
---
в скд это куда предлагаете записать?
---
В запрос например ;)
---
круть, вот так прямо по серед двора ? (запроса)
---
может запрос покажете?
---
```1c
    ВЫБРАТЬ
    СчетНаОплатуПокупателюТовары.Номенклатура КАК НоменклатураЗаказано,
    СчетНаОплатуПокупателю.Ссылка КАК СчетНаОплатуКлиенту,
    СчетНаОплатуПокупателюТовары.Номенклатура.Артикул КАК АртикулЗаказаннойНоменклатуры,
    СУММА(СчетНаОплатуПокупателюТовары.Количество) КАК КоличествоЗаказано,
    СчетНаОплатуПокупателю.Контрагент.Ссылка КАК Клиент
    ПОМЕСТИТЬ Вт_СчетНаОплатуПокупателю
    ИЗ
    Документ.СчетНаОплатуПокупателю КАК СчетНаОплатуПокупателю
    ПОЛНОЕ СОЕДИНЕНИЕ Документ.СчетНаОплатуПокупателю.Товары КАК СчетНаОплатуПокупателюТовары
    ПО СчетНаОплатуПокупателю.Ссылка = СчетНаОплатуПокупателюТовары.Ссылка
    ГДЕ
    СчетНаОплатуПокупателю.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И СчетНаОплатуПокупателю.ПометкаУдаления = ЛОЖЬ
    И СчетНаОплатуПокупателю.Проведен = ИСТИНА

    СГРУППИРОВАТЬ ПО
    СчетНаОплатуПокупателюТовары.Номенклатура,
    СчетНаОплатуПокупателю.Ссылка,
    СчетНаОплатуПокупателюТовары.Номенклатура.Артикул,
    СчетНаОплатуПокупателю.Контрагент.Ссылка
    ;

    ////////////////////////////////////////////////////////////­­////////////////////
    ВЫБРАТЬ
    РеализацияТоваровУслугТовары.Номенклатура КАК НоменклатураРеализовано,
    СУММА(РеализацияТоваровУслугТовары.Количество) КАК КоличествоРеализовано,
    РеализацияТоваровУслуг.СчетНаОплатуПокупателю КАК РеализованоСчетНаОплатуПокупателю
    ПОМЕСТИТЬ Вт_РеализованоКлиенту
    ИЗ
    Документ.РеализацияТоваровУслуг КАК РеализацияТоваровУслуг
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    ПО РеализацияТоваровУслуг.Ссылка = РеализацияТоваровУслугТовары.Ссылка

    СГРУППИРОВАТЬ ПО
    РеализацияТоваровУслуг.СчетНаОплатуПокупателю,
    РеализацияТоваровУслугТовары.Номенклатура
    ;

    ////////////////////////////////////////////////////////////­­////////////////////
    ВЫБРАТЬ
    Вт_СчетНаОплатуПокупателю.СчетНаОплатуКлиенту КАК СчетНаОплатуКлиенту,
    Вт_СчетНаОплатуПокупателю.НоменклатураЗаказано КАК НоменклатураЗаказано,
    Вт_СчетНаОплатуПокупателю.АртикулЗаказаннойНоменклатуры КАК АртикулЗаказаннойНоменклатуры,
    Вт_СчетНаОплатуПокупателю.КоличествоЗаказано КАК КоличествоЗаказано,
    Вт_СчетНаОплатуПокупателю.Клиент КАК Клиент,
    Вт_РеализованоКлиенту.НоменклатураРеализовано КАК НоменклатураРеализовано,
    ВЫБОР
    КОГДА Вт_РеализованоКлиенту.КоличествоРеализовано ЕСТЬ NULL
    ТОГДА 0
    ИНАЧЕ Вт_РеализованоКлиенту.КоличествоРеализовано
    КОНЕЦ КАК КоличествоРеализовано,
    Вт_СчетНаОплатуПокупателю.КоличествоЗаказано - Вт_РеализованоКлиенту.КоличествоРеализовано КАК Остаток
    ИЗ
    Вт_СчетНаОплатуПокупателю КАК Вт_СчетНаОплатуПокупателю
    ЛЕВОЕ СОЕДИНЕНИЕ Вт_РеализованоКлиенту КАК Вт_РеализованоКлиенту
    ПО Вт_СчетНаОплатуПокупателю.НоменклатураЗаказано.Ссылка = Вт_РеализованоКлиенту.НоменклатураРеализовано.Ссылка
    И Вт_СчетНаОплатуПокупателю.СчетНаОплатуКлиенту.Ссылка = Вт_РеализованоКлиенту.РеализованоСчетНаОплатуПокупателю.Ссылка
```
---
Не боись, я так сто раз делал))
---
как такая реализация кода поможет исключить те заказы ,в которых количество заказанных товаров = реализовано?
---
```1c
    ВЫБРАТЬ
    СчетНаОплатуПокупателюТовары.Номенклатура КАК НоменклатураЗаказано,
    СчетНаОплатуПокупателю.Ссылка КАК СчетНаОплатуКлиенту,
    СчетНаОплатуПокупателюТовары.Номенклатура.Артикул КАК АртикулЗаказаннойНоменклатуры,
    СУММА(СчетНаОплатуПокупателюТовары.Количество) КАК КоличествоЗаказано,
    СчетНаОплатуПокупателю.Контрагент.Ссылка КАК Клиент
    ПОМЕСТИТЬ Вт_СчетНаОплатуПокупателю
    ИЗ
    Документ.СчетНаОплатуПокупателю КАК СчетНаОплатуПокупателю
    ПОЛНОЕ СОЕДИНЕНИЕ Документ.СчетНаОплатуПокупателю.Товары КАК СчетНаОплатуПокупателюТовары
    ПО СчетНаОплатуПокупателю.Ссылка = СчетНаОплатуПокупателюТовары.Ссылка
    ГДЕ
    СчетНаОплатуПокупателю.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И СчетНаОплатуПокупателю.ПометкаУдаления = ЛОЖЬ
    И СчетНаОплатуПокупателю.Проведен = ИСТИНА

    СГРУППИРОВАТЬ ПО
    СчетНаОплатуПокупателюТовары.Номенклатура,
    СчетНаОплатуПокупателю.Ссылка,
    СчетНаОплатуПокупателюТовары.Номенклатура.Артикул,
    СчетНаОплатуПокупателю.Контрагент.Ссылка
    ;

    ////////////////////////////////////////////////////////////­­­////////////////////
    ВЫБРАТЬ
    РеализацияТоваровУслугТовары.Номенклатура КАК НоменклатураРеализовано,
    СУММА(РеализацияТоваровУслугТовары.Количество) КАК КоличествоРеализовано,
    РеализацияТоваровУслуг.СчетНаОплатуПокупателю КАК РеализованоСчетНаОплатуПокупателю
    ПОМЕСТИТЬ Вт_РеализованоКлиенту
    ИЗ
    Документ.РеализацияТоваровУслуг КАК РеализацияТоваровУслуг
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    ПО РеализацияТоваровУслуг.Ссылка = РеализацияТоваровУслугТовары.Ссылка

    СГРУППИРОВАТЬ ПО
    РеализацияТоваровУслуг.СчетНаОплатуПокупателю,
    РеализацияТоваровУслугТовары.Номенклатура
    ;

    ////////////////////////////////////////////////////////////­­­////////////////////
    ВЫБРАТЬ
    Вт_СчетНаОплатуПокупателю.СчетНаОплатуКлиенту КАК СчетНаОплатуКлиенту,
    Вт_СчетНаОплатуПокупателю.НоменклатураЗаказано КАК НоменклатураЗаказано,
    Вт_СчетНаОплатуПокупателю.АртикулЗаказаннойНоменклатуры КАК АртикулЗаказаннойНоменклатуры,
    Вт_СчетНаОплатуПокупателю.КоличествоЗаказано КАК КоличествоЗаказано,
    Вт_СчетНаОплатуПокупателю.Клиент КАК Клиент,
    Вт_РеализованоКлиенту.НоменклатураРеализовано КАК НоменклатураРеализовано,
    ВЫБОР
    КОГДА Вт_РеализованоКлиенту.КоличествоРеализовано ЕСТЬ NULL
    ТОГДА 0
    ИНАЧЕ Вт_РеализованоКлиенту.КоличествоРеализовано
    КОНЕЦ КАК КоличествоРеализовано,
    Вт_СчетНаОплатуПокупателю.КоличествоЗаказано - Вт_РеализованоКлиенту.КоличествоРеализовано КАК Остаток
    ИЗ
    Вт_СчетНаОплатуПокупателю КАК Вт_СчетНаОплатуПокупателю
    ЛЕВОЕ СОЕДИНЕНИЕ Вт_РеализованоКлиенту КАК Вт_РеализованоКлиенту
    ПО Вт_СчетНаОплатуПокупателю.НоменклатураЗаказано.Ссылка = Вт_РеализованоКлиенту.НоменклатураРеализовано.Ссылка
    И Вт_СчетНаОплатуПокупателю.СчетНаОплатуКлиенту.Ссылка = Вт_РеализованоКлиенту.РеализованоСчетНаОплатуПокупателю.Ссылка
    ГДЕ
    ВЫБОР КОГДА &УбратьРеализованные ТОГДА
    Вт_СчетНаОплатуПокупателю.КоличествоЗаказано <> ЕСТЬNULL(Вт_РеализованоКлиенту.КоличествоРеализовано,0)
    ИНАЧЕ

    ИСТИНА
    КОНЕЦ
```
---
у меня мысль была иначе.
если количество заказанных товаров = реализовано значит Остаток "Ноль".

и тут на форме (скрин) если я отмечаю "ОтсевОтгружденых" - "Да" (Истина).
то те заказы, в которых Остаток "Ноль" - исчезают из отчета.
---
т на форме (скрин) если я отмечаю "ОтсевОтгружденых" - "Да" (Истина).
то те заказы, в которых Остаток "Ноль" - исчезают из отчета.

Так мой вариант и предполагает что если ОтсевОтгруженных = Да то мы убираем из запроса строки где количество из заказа = количество реализовано
---
согласен, таки да.